Grafy funkcí

<u>zadaní:</u> vyšetřete grafy funkcí

1) f = $\frac{x^2-1}{x^2+1}$

řešení

```
> f:=x->(x^2-1)/(x^2+1);
```

$$f := x \to \frac{x^2-1}{x^2+1}$$

funkce **f** je sudá (tzn. f(x)=f(-x)), spojitá na ($-\infty$, ∞) a omezená $|\boldsymbol{f(x)}| \le \boldsymbol{1}$

```
> abs(f(x))<=1;
```

$$\left|\frac{x^2-1}{1+x^2}\right| \le 1$$

limity v krajních bodech (krajní body této funkce **f** jsou ∞ a $-\infty$) :

$\lim_{x \to \infty} f(x) = 1$ a $\lim_{x \to (-\infty)} f(x) = 1$

```
> Limit(f(x),x=infinity)=limit(f(x),x=infinity);
```

$$\lim_{x \to \infty} \frac{x^2-1}{1+x^2} = 1$$

```
> Limit(f(x),x=-infinity)=limit(f(x),x=-infinity);
```

$$\lim_{x \to (-\infty)} \frac{x^2-1}{1+x^2} = 1$$

monotonie funkce f :

```
> deriv1:=D(f);
```

$$deriv1 := x \to \frac{2\,x}{x^2+1} - \frac{2\,(x^2-1)\,x}{(x^2+1)^2}$$

```
> deriv1:=unapply(simplify(deriv1(x)),x);
```

$$deriv1 := x \to \frac{4\,x}{(x^2+1)^2}$$

derivace funkce f je pro kladná x <u>kladná</u> => f je **rostoucí** pro kladná x
derivace funkce f je pro kladná x <u>záporná</u> => f je **klesající** pro záporná x
v bodě <u>x = 0</u> má funkce f **minimum** a f(0)= -1:

```
> y:=f(0);
```

$$y := -1$$

obor hodnot : H(f) = <-1,1>

konkávnita a **konvexita** funkce f : udělám druhou derivaci funkce f

```
> deriv2:=D(deriv1);
```

$$deriv2 := x \to \frac{4}{(x^2+1)^2} - \frac{16\,x^2}{(x^2+1)^3}$$

```
> deriv2:=simplify(deriv2(x));
```

$$deriv2 := -\frac{4\,(-1+3\,x^2)}{(1+x^2)^3}$$

```
> solve(deriv2=0,x);
```

$$-\frac{\sqrt{3}}{3},\ \frac{\sqrt{3}}{3}$$

```
> solve(deriv2<0,x);
```

$$\mathrm{RealRange}\left(-\infty, \mathrm{Open}\left(-\frac{\sqrt{3}}{3}\right)\right),\ \mathrm{RealRange}\left(\mathrm{Open}\left(\frac{\sqrt{3}}{3}\right), \infty\right)$$

```
> solve(deriv2>0,x);
```

$$\mathrm{RealRange}\left(\mathrm{Open}\left(-\frac{\sqrt{3}}{3}\right), \mathrm{Open}\left(\frac{\sqrt{3}}{3}\right)\right)$$

funkce f je **konvexní** na intervalu ($-\frac{\sqrt{3}}{3}$, $\frac{\sqrt{3}}{3}$) - druhá derivace je na tomto intervalu <u>kladná</u>

konkávní na intervalu ($-\infty$, $-\frac{\sqrt{3}}{3}$) a na intervalu ($\frac{\sqrt{3}}{3}$, ∞) - druhá derivace je na těchto intervalech <u>záporná</u>

inflexní body : $-\frac{\sqrt{3}}{3}$ a $\frac{\sqrt{3}}{3}$

graf funkce **f** = $\frac{x^2-1}{x^2+1}$ je vyznačen červeně (její omezení shora je nakresleno modře a modře jsou též vyznačeny inflexní body) :

```
> plot1:=plot(f(x),x=-10..10,-1.5..1.5,color=red):
> plot2:=plot([1,[[-1/sqrt(3),f(-1/sqrt(3))],[1/sqrt(3),f(1/sqrt(3))]]], -10..10,style=[line,point],color=blue):
> plots[display]([plot1,plot2]);
```



2) f = $\ln\left(\mathbf{e}+\frac{1}{x}\right)$

řešení

```
> f:=x->ln(exp(1)+1/x);
```

$$f := x \rightarrow \ln\left(\mathbf{e}+\frac{1}{x}\right)$$

Definiční obor :

```
> solve((exp(1)+1/x)>0,x);
```

$$\mathrm{RealRange}\left(-\infty, \mathrm{Open}\left(-\frac{1}{\mathbf{e}}\right)\right), \mathrm{RealRange}(\mathrm{Open}(0), \infty)$$

Definičním oborem funkce f je sjednocení intervalů $(-\infty , -\frac{1}{\mathbf{e}})$ a $(0 , \infty)$

Spojitost : f je spojitá na celém svém definičním oboru

Limity v krajních bodech tj. $-\infty , -\frac{1}{\mathbf{e}} , 0 , \infty$

$\lim_{x \to (-\infty)} f(x) = 1$, $\lim_{x \to \infty} f(x) = 1$, $\lim_{x \to \left(-\frac{1}{\mathbf{e}}\right)-} f(x) = -\infty$, $\lim_{x \to 0+} f(x) = \infty$

```
> Limit(f(x),x=-infinity)=limit(f(x),x=-infinity);
```

$$\lim_{x \to (-\infty)} \ln\left(\mathbf{e} + \frac{1}{x}\right) = 1$$

```
> Limit(f(x),x=infinity)=limit(f(x),x=infinity);
```

$$\lim_{x \to \infty} \ln\left(\mathbf{e} + \frac{1}{x}\right) = 1$$

```
> Limit(f(x),x=-1/exp(1),left)=limit(f(x),x=-1/exp(1),left);
```

$$\lim_{x \to \left(-\frac{1}{\mathbf{e}}\right)-} \ln\left(\mathbf{e} + \frac{1}{x}\right) = -\infty$$

```
> Limit(f(x),x=0,right)=limit(f(x),x=0,right);
```

$$\lim_{x \to 0+} \ln\left(\mathbf{e} + \frac{1}{x}\right) = \infty$$

Obor hodnot : H(f) = R \ {1}

monotonie funkce f : stačí zjistit, na kterých intervalech je derivace funkce kladná a na kterých záporná

```
> deriv1:=D(f);
```

$$deriv1 := x \to -\frac{1}{x^2\left(\mathbf{e} + \frac{1}{x}\right)}$$

```
> solve(deriv1(x)<0,x);
```

$$\mathrm{RealRange}\left(-\infty, \mathrm{Open}\left(-\frac{1}{\mathbf{e}}\right)\right), \mathrm{RealRange}(\mathrm{Open}(0), \infty)$$

Funkce f je **klesající** na <u>celém</u> svém definičním oboru (první derivace funkce f je zde záporná)

konvexita a **konkávnita** funkce f : stačí zjistit , na kterých intervalech je druhá derivace funkce kladná a na kterých záporná

```
> deriv2:=D(deriv1);
```

$$deriv2 := x \to \frac{2}{x^3\left(\mathbf{e} + \frac{1}{x}\right)} - \frac{1}{x^4\left(\mathbf{e} + \frac{1}{x}\right)^2}$$

```
> deriv2:=simplify(deriv2(x));
```

$$deriv2 := \frac{2\,\mathbf{e}\,x + 1}{x^2\,(\mathbf{e}\,x + 1)^2}$$

```
> solve(deriv2>0,x);
```

$$\mathrm{RealRange}\left(\mathrm{Open}\left(-\frac{1}{2}\frac{1}{\mathbf{e}}\right), \mathrm{Open}(0)\right), \mathrm{RealRange}(\mathrm{Open}(0), \infty)$$

```
> solve(deriv2<0,x);
```

$$\mathrm{RealRange}\left(-\infty, \mathrm{Open}\left(-\frac{1}{\mathbf{e}}\right)\right), \mathrm{RealRange}\left(\mathrm{Open}\left(-\frac{1}{\mathbf{e}}\right), \mathrm{Open}\left(-\frac{1}{2}\frac{1}{\mathbf{e}}\right)\right)$$

Funkce f je **konvexní** na intervalu (0 , ∞) - na tomto intervalu (část definičního oboru) je druhá derivace kladná

konkávní na intervalu ($-\infty$, $-\frac{1}{\mathbf{e}}$) - na tomto intervalu (část definičního oboru) je druhá derivace záporná

Graf funkce **f** **=** $\ln\left(\mathbf{e}+\frac{1}{x}\right)$ je vyznačen červeně (pro názornost jsou zeleně vyznačeny funkce y=1 a x= $-\frac{1}{\mathbf{e}}$).

```
> plot1:=plot(f(x),x=-3..4,-4..3.5,color=red):
> plot2 :=
  plots[textplot]([-exp(-1)-0.01,0.01,`-exp(-1)`],align=ABOVE):
> plot3:=plot([1,[[-exp(-1),-4],[-exp(-1),3.5]]],-3..4,-4..3.5,
  color=green):
> plots[display]([plot1,plot2,plot3]);
```



>
>